Trabalho 2: Modelagem e Análise de Dados Legislativos com Neo4j

Banco de Dados para Ciência de Dados

Guilherme César Athayde - 748175 guilherme.athayde@estudante.ufscar.br

Nome do Segundo Autor - RA email@estudante.ufscar.br

23 de outubro de 2025

Sumário

1	Intr	odução 3	
	1.1	Dataset Escolhido	
	1.2	Características Importantes do Dataset	
2	Col	eta de Dados	
	2.1	Script de Coleta	
3	Mo	delagem dos Dados 4	
	3.1	Entidades (Nós)	
	3.2	Relacionamentos (Arestas)	
4	Imp	ortação dos Dados 6	
	4.1	Script de Importação	
5	Consulta de Verificação 7		
	5.1	Script da Consulta	
	5.2	Resultado da Consulta	
6	Consultas Analíticas e Insights 8		
	6.1	Consulta 1: Presença Partidária em Votações	
	6.2	Consulta 2: Impacto de Proposições Formais	
	6.3	Consulta 3: Análise Semântica de Descrições	
	6.4	Consulta 4: Volatilidade de Resultados Partidários	
	6.5	Consulta 5: Deputados Mais Influentes	
7	Cor	iclusão 11	
	7.1	Aprendizados	
	7.2	Desafios Encontrados	
	7.3	Repositório do Projeto	

1 Introdução

Este trabalho apresenta a modelagem, importação e análise de dados legislativos da Câmara dos Deputados do Brasil utilizando o banco de dados de grafos Neo4j. O objetivo é explorar as relações entre deputados, partidos, proposições e votações para gerar insights sobre o funcionamento do processo legislativo brasileiro.

1.1 Dataset Escolhido

O dataset utilizado foi coletado da API de Dados Abertos da Câmara dos Deputados (https://dadosabertos.camara.leg.br), contendo informações sobre:

- **Deputados**: 513 registros contendo informações sobre parlamentares da 57^a Legislatura (2023-2027), incluindo nome, partido, UF, email e foto.
- Partidos: 28 partidos políticos com representação na Câmara, contendo sigla, nome completo e URI de referência.
- **Proposições**: 15 proposições legislativas (PLs, PECs, etc.) com ementa, tipo, número e ano.
- Votações: 100 registros de votações realizadas em diferentes órgãos (Plenário, comissões) com descrição, resultado (aprovado/rejeitado) e data.
- Órgãos: 8 órgãos legislativos incluindo o Plenário e diversas comissões temáticas.
- Frentes Parlamentares: 15 frentes parlamentares temáticas da legislatura atual.

Os dados referem-se principalmente ao período de 2023-2025, com foco na legislatura atual e votações recentes (outubro de 2025).

1.2 Características Importantes do Dataset

- Os dados são estruturados em formato JSON seguindo o padrão da API REST da Câmara.
- Existem relacionamentos naturais entre entidades: deputados pertencem a partidos, votações são realizadas por órgãos, proposições podem estar vinculadas a votações.
- As descrições de votações frequentemente mencionam nomes de deputados (relatores, autores), permitindo análises textuais.
- O campo aprovacao nas votações indica o resultado (1=aprovado, 0=rejeitado, null=outros).

2 Coleta de Dados

Os dados foram coletados através de requisições HTTP à API de Dados Abertos da Câmara dos Deputados. O script de coleta utiliza curl para fazer as requisições e jq para formatar o JSON de saída.

2.1 Script de Coleta

Listing 1: Script de coleta de dados


```
cd datasets
  curl -s -H "Content-Type: application/json" \
    https://dadosabertos.camara.leg.br/api/v2/deputados | jq '.' >
        deputados.json
4
  curl -s -H "Content-Type: application/json" \
5
    https://dadosabertos.camara.leg.br/api/v2/proposicoes | jq '.' >
6
        proposicoes.json
  curl -s -H "Content-Type: application/json" -H "Accept: application/json
    https://dadosabertos.camara.leg.br/api/v2/votacoes | jq '.' > votacoes
9
        .json
  curl -s -H "Content-Type: application/json" \
    https://dadosabertos.camara.leg.br/api/v2/orgaos | jq '.' > orgaos.
        json
  curl -s -H "Content-Type: application/json" \
14
    https://dadosabertos.camara.leg.br/api/v2/partidos | jq '.' > partidos
        .json
  curl -s -H "Content-Type: application/json" \
17
    https://dadosabertos.camara.leg.br/api/v2/frentes | jq '.' > frentes.
18
        json
```

3 Modelagem dos Dados

A modelagem dos dados em grafo foi projetada para representar as relações naturais entre as entidades do processo legislativo. O diagrama abaixo ilustra o modelo de dados implementado no Neo4j:

3.1 Entidades (Nós)

- Partido: Partidos políticos com sigla, nome e URI.
- Deputado: Parlamentares com id, nome, partido, UF, legislatura, email e foto.
- Legislatura: Períodos legislativos identificados por ID.
- Frente: Frentes parlamentares temáticas.
- Orgao: Órgãos legislativos (Plenário, comissões).
- Proposição: Proposições legislativas com tipo, número, ano e ementa.
- Votacao: Eventos de votação com data, descrição e resultado.






Figura 1: Diagrama do modelo de dados em grafo

3.2 Relacionamentos (Arestas)

- PERTENCE_A: Deputado → Partido
- ATUA_NA_LEGISLATURA: Deputado \rightarrow Legislatura
- DA_LEGISLATURA: Frente \rightarrow Legislatura
- REALIZA: Orgao → Votacao
- REFERENCIADA_EM: Proposicao → Votacao

4 Importação dos Dados

A importação dos dados JSON para o Neo4j foi realizada através de um script Python que utiliza o driver oficial do Neo4j. O script implementa as seguintes funcionalidades:

- Leitura dos arquivos JSON do diretório datasets/
- Criação de constraints de unicidade para cada entidade
- Merge (criação ou atualização) dos nós
- Criação dos relacionamentos entre nós
- Tratamento de valores nulos para evitar poluição do grafo

4.1 Script de Importação

Listing 2: Trecho principal do script de importação (import_neo4j.py)

```
def merge_deputados(driver: Driver, database: str,
                        deputados: List[Dict[str, Any]]):
2
       query = """
3
       UNWIND $rows AS row
4
       MERGE (d:Deputado {id: row.id})
         ON CREATE SET d.nome = row.nome,
6
                        d.siglaPartido = row.siglaPartido,
                        d.siglaUf = row.siglaUf,
                        d.idLegislatura = row.idLegislatura,
                        d.email = row.email,
10
                        d.urlFoto = row.urlFoto,
11
                        d.uri = row.uri
12
         ON MATCH SET
                        d.nome = row.nome,
13
                        d.siglaPartido = row.siglaPartido,
14
                        d.siglaUf = row.siglaUf,
                        d.idLegislatura = row.idLegislatura,
16
                        d.email = row.email,
17
                        d.urlFoto = row.urlFoto,
18
                        d.uri = row.uri
19
       WITH d, row
20
       MATCH (p:Partido {sigla: row.siglaPartido})
21
       MERGE (p) <-[:PERTENCE_A]-(d)
22
       WITH d, row
23
       MATCH (1:Legislatura {id: row.idLegislatura})
```

```
MERGE (d) - [: ATUA_NA_LEGISLATURA] -> (1)
25
26
       driver.execute_query(query, rows=deputados, database_=database)
27
28
   # Processo principal de importacao
29
   print("Conectado ao Neo4j. Criando constraints...")
30
   create_constraints(driver, args.database)
31
   print("Importando Partidos...")
32
  merge_partidos(driver, args.database, partidos)
  print("Importando Legislaturas...")
34
  merge_legislaturas(driver, args.database, legislatura_ids)
35
  print("Importando Deputados...")
36
  merge_deputados(driver, args.database, deputados)
  print("Importando Frentes...")
38
  merge_frentes(driver, args.database, frentes)
39
  print("Importando Orgaos...")
40
  merge_orgaos(driver, args.database, orgaos)
  print("Importando Proposicoes...")
42
  merge_proposicoes(driver, args.database, proposicoes)
43
  print("Importando Votacoes...")
44
  merge_votacoes(driver, args.database, votacoes)
  print("Importacao concluida.")
```

5 Consulta de Verificação

Após a importação, foi realizada uma consulta para verificar todos os dados importados no banco de dados Neo4j.

5.1 Script da Consulta

Listing 3: Consulta de verificação em main.ipynb

```
from neo4j import GraphDatabase
  URI = "neo4j+s://ae951e8f.databases.neo4j.io"
  USERNAME = "neo4j"
4
  PASSWORD = "E_shPCUTJVDIZmj-iWC-p2skPNyrPBcT148nZyh9BgA"
5
  driver = GraphDatabase.driver(URI, auth=(USERNAME, PASSWORD))
  driver.verify_connectivity()
8
9
  query = """
10
  OPTIONAL MATCH (n)-[r]->(m)
11
   RETURN n, r, m
12
14
  records, summary, keys = driver.execute_query(query, database_="neo4j")
16
  for i, record in enumerate(records):
17
       node_n = record["n"]
       relationship_r = record["r"]
19
       node_m = record["m"]
20
21
       print(f"\nRegistro {i+1}:")
       if node_n:
23
```

```
print(f" No de Origem (n): Labels={list(node_n.labels)}, "
24
                  f "Propriedades = {dict(node_n)}")
25
       if relationship_r:
26
           print(f" Relacao (r): Tipo='{relationship_r.type}', "
27
                  f "Propriedades = { dict(relationship_r)}")
       if node_m:
29
                    No de Destino (m): Labels={list(node_m.labels)}, "
30
                  f "Propriedades = {dict(node_m)}")
31
32
  driver.close()
```

5.2 Resultado da Consulta

A consulta retornou 476 registros de relacionamentos entre nós, confirmando a importação bem-sucedida de:

- 513 deputados vinculados a partidos e legislaturas
- 28 partidos
- 1 legislatura (ID 57)
- 15 frentes parlamentares
- 8 órgãos
- 15 proposições
- 100 votações

O resultado completo encontra-se no arquivo main.txt, mostrando todos os nós, relacionamentos e suas propriedades.

6 Consultas Analíticas e Insights

Foram desenvolvidas 5 consultas Cypher avançadas para extrair insights sobre o funcionamento do processo legislativo. As consultas exploram diferentes dimensões do grafo, incluindo análises partidárias, comportamento de votações e identificação de atores influentes.

6.1 Consulta 1: Presença Partidária em Votações

Pergunta: Quais partidos aparecem mais frequentemente citados nas descrições de votações (indicando atuação de seus deputados como relatores ou autores) e essa presença está associada a maior taxa de aprovação?

Listing 4: Consulta Cypher 1

```
MATCH (p:Partido) <-[:PERTENCE_A]-(d:Deputado)

MATCH (o:Orgao)-[:REALIZA]->(v:Votacao)

WHERE v.descricao CONTAINS d.nome

WITH p, o, v

WITH p.sigla AS partido, o.sigla AS orgao,
```

```
collect(DISTINCT v.id) AS votacoes_ids,
count(DISTINCT v) AS qtd_votacoes,
avg(CASE WHEN v.aprovacao = 1 THEN 1.0 ELSE 0.0 END) AS taxa_aprov

RETURN partido, orgao, qtd_votacoes,
round(100.0 * taxa_aprov,2) AS taxaAprovacaoPercent,
votacoes_ids

ORDER BY qtd_votacoes DESC, taxaAprovacaoPercent DESC
```

Análise: Esta consulta aproxima a "visibilidade" de um partido nas votações via menções diretas a nomes de deputados em descrições oficiais (frequentemente indicando relatoria ou protagonismo). Partidos com alto número de menções em conjunto com elevada taxa de aprovação sugerem capacidade de pautar e conduzir matérias a resultados positivos, refletindo coordenação interna ou posicionamento estratégico em comissões e plenário. Já partidos muito citados com baixa taxa de aprovação podem estar assumindo temas controversos, atuando em obstrução ou em fases intermediárias (ex.: emendas rejeitadas). O insight auxilia na identificação de quais siglas convertem presença em efetividade decisória.

6.2 Consulta 2: Impacto de Proposições Formais

Pergunta: Votações associadas explicitamente a uma proposição têm maior taxa de aprovação em cada órgão do que votações sem essa referência?

Listing 5: Consulta Cypher 2

```
MATCH (o:Orgao) - [:REALIZA] -> (v:Votacao)
   OPTIONAL MATCH (p:Proposicao) - [:REFERENCIADA_EM] -> (v)
2
  WITH o, v, CASE WHEN p IS NULL THEN O ELSE 1 END AS temProposicao
  WITH o.sigla AS orgao,
4
        count(v) AS total,
5
        sum(temProposicao) AS comProp,
6
        avg(CASE WHEN v.aprovacao = 1 THEN 1.0 ELSE 0.0 END) AS taxa_geral,
        avg(CASE WHEN temProposicao = 1 AND v.aprovacao = 1
8
             THEN 1.0 ELSE 0.0 END) AS taxa_com_prop,
9
        avg(CASE WHEN temProposicao = 0 AND v.aprovacao = 1
             THEN 1.0 ELSE 0.0 END) AS taxa_sem_prop
  RETURN orgao, total,
12
          comProp AS votacoesComProposicao,
          round(100.0 * comProp / total,2) AS percComProposicao,
14
          round(100.0 * taxa_geral,2) AS taxaAprovGeral,
          round(100.0 * taxa_com_prop,2) AS taxaAprovComProposicao,
16
          round(100.0 * taxa_sem_prop,2) AS taxaAprovSemProposicao
17
  ORDER BY total DESC
```

Análise: A diferença entre taxa de aprovação com e sem proposição vinculada revela o papel do rito formal na previsibilidade do resultado. Órgãos em que a aprovação depende fortemente da presença de uma proposição podem funcionar como estágios de consolidação técnica/jurídica antes de avançar para debates mais abertos. Um gap grande sugere que votações "avulsas" (ex.: requerimentos administrativos) são mais sujeitas a rejeição ou controvérsia.

6.3 Consulta 3: Análise Semântica de Descrições

Pergunta: Certos termos ("Redação Final", "Parecer", "Substitutivo", "Requerimento", "Emenda") nas descrições estão ligados a taxas de aprovação distintas?

Listing 6: Consulta Cypher 3

```
MATCH (o:Orgao) -[:REALIZA] ->(v:Votacao)
   WITH o, v,
2
        [t IN ['Redacao Final', 'Parecer', 'Substitutivo',
               'Requerimento', 'Emenda']
4
         WHERE v.descricao CONTAINS t] AS termos,
        (v.aprovacao = 1) AS aprovado
6
  UNWIND termos AS termo
   WITH o.sigla AS orgao, termo, aprovado
8
  RETURN orgao, termo,
9
          count(*) AS ocorrencias,
10
          sum (CASE WHEN aprovado THEN 1 ELSE 0 END) AS aprovadas,
          round (100.0 * sum (CASE WHEN aprovado THEN 1 ELSE 0 END)
12
                /count(*),2) AS taxaAprovacaoTermo
  ORDER BY orgao, ocorrencias DESC
14
```

Análise: Palavras como "Redação Final" e "Parecer" tendem a representar fases conclusivas do processo legislativo, frequentemente após negociações e ajustes; altas taxas de aprovação nesses casos refletem maturidade e consenso já construído. "Emenda" ou "Substitutivo" podem carregar maior volatilidade: rejeições de emendas indicam disputas sobre mérito ou estratégia. Comparar órgãos mostra onde o texto técnico (Parecer) tem mais peso (ex.: comissões temáticas) e onde expressões processuais (Requerimento) mantêm alta aprovação.

6.4 Consulta 4: Volatilidade de Resultados Partidários

Pergunta: A atuação textual (menções) de partidos em votações apresenta padrões de estabilidade (resultados quase sempre aprovados) ou alta variância?

Listing 7: Consulta Cypher 4

```
MATCH (p:Partido) <-[:PERTENCE_A]-(d:Deputado)</pre>
  MATCH (o:Orgao) -[:REALIZA] ->(v:Votacao)
2
   WHERE v.descricao CONTAINS d.nome
   WITH p, v
4
   WITH p.sigla AS partido,
5
        collect (v.aprovacao) AS resultados,
        avg(CASE WHEN v.aprovacao = 1 THEN 1.0 ELSE 0.0 END) AS taxa,
7
        stDev(toFloat(CASE WHEN v.aprovacao IS NULL THEN 0
8
                            ELSE v.aprovacao END)) AS desvio
9
  RETURN partido,
          size(resultados) AS qtdCitacoes,
11
          round(100.0 * taxa,2) AS taxaAprovacaoPercent,
          round(desvio,3) AS desvioPadraoAprovacao,
          resultados
  ORDER BY qtdCitacoes DESC
```

Análise: O desvio-padrão quantifica volatilidade dos resultados em votações onde o partido está textual/semanticamente presente. Partidos com baixa variância e alta taxa de aprovação sugerem foco em matérias de consenso ou eficiência em negociação. Alta variância pode revelar estratégia de presença em pautas polarizadas (aceitando risco), tentativa de moldar agenda ou atuação defensiva.

6.5 Consulta 5: Deputados Mais Influentes

Pergunta: Quais deputados aparecem com maior frequência nas descrições de votações e essa visibilidade está correlacionada a maior taxa de aprovação?

Listing 8: Consulta Cypher 5

```
MATCH (d:Deputado)
   MATCH (o:Orgao) -[:REALIZA] ->(v:Votacao)
   WHERE v.descricao CONTAINS d.nome
3
  WITH d, v
4
   WITH d.nome AS deputado,
        d.siglaPartido AS partido,
6
        d.siglaUf AS uf,
        count(DISTINCT v) AS citacoes,
        avg(CASE WHEN v.aprovacao = 1 THEN 1.0 ELSE 0.0 END) AS taxa
9
  RETURN deputado, partido, uf, citacoes,
          round(100.0 * taxa,2) AS taxaAprovacaoPercent
11
  ORDER BY citacoes DESC
12
  LIMIT 25
```

Análise: Deputados frequentemente citados podem ocupar funções de relatoria, liderança de bancada ou protagonismo em áreas temáticas. Uma alta taxa de aprovação nas votações onde aparecem sugere capacidade de construir acordos e navegar ritos procedimentais; já citações numerosas com taxa reduzida podem refletir exposição em pautas contestadas ou papel de oposicionista. O ranking auxilia na identificação de "influenciadores processuais".

7 Conclusão

7.1 Aprendizados

Este trabalho proporcionou diversos aprendizados importantes sobre modelagem e análise de dados em grafos:

- Modelagem Natural: O Neo4j permitiu uma representação natural e intuitiva das relações entre entidades legislativas, capturando a complexidade das interações políticas de forma mais expressiva que modelos relacionais tradicionais.
- Poder do Cypher: A linguagem Cypher demonstrou-se muito expressiva para consultas complexas, permitindo traversals de múltiplos níveis, agregações sofisticadas e análises estatísticas (como desvio-padrão) de forma declarativa.
- Análise Textual: A capacidade de combinar análise estrutural (grafos) com busca textual (CONTAINS) permitiu inferir relacionamentos implícitos (relatoria, autoria) a partir de descrições não estruturadas.
- Visualização de Relações: O modelo de grafo facilitou a identificação de padrões de influência e redes de colaboração que seriam difíceis de detectar em modelos tabulares.

7.2 Desafios Encontrados

Durante o desenvolvimento do projeto, foram enfrentados os seguintes desafios:

- Qualidade dos Dados: A API retorna dados limitados (primeiras páginas), exigindo adaptação do escopo da análise. Alguns campos contêm valores nulos ou inconsistentes.
- Relacionamentos Implícitos: A ausência de dados explícitos de autoria e votação nominal limitou a análise, sendo necessário recorrer a heurísticas textuais (busca de nomes em descrições), o que pode gerar falsos positivos.
- **Desempenho**: Consultas com **CONTAINS** em grandes volumes de texto podem ser lentas. A criação de índices full-text seria necessária para escalabilidade.
- Normalização de Nomes: Variações de grafia, abreviações e homônimos nos nomes de deputados podem afetar a precisão das análises textuais.

7.3 Repositório do Projeto

O código-fonte completo, datasets e documentação estão disponíveis no repositório GitHub:

https://github.com/guiathayde/neo4j-camara-deputados

Referências

- Câmara dos Deputados. API de Dados Abertos. Disponível em: https://dadosabertos.camara.leg.br/. Acesso em: 16 out. 2025.
- Neo4j. Cypher Manual. Disponível em: https://neo4j.com/docs/cypher-manual/. Acesso em: 16 out. 2025.
- Robinson, I., Webber, J., & Eifrem, E. (2015). *Graph Databases: New Opportunities for Connected Data*. O'Reilly Media.